
//*********************************************
//
// NAME: WPCH_IN1.WPM
//
// PURPOSE: Coach on Tab and Indent
//
// AUTHOR: Paul Laing, WordPerfect Corp.
//
//*********************************************

// Revision History
//
//

Global(ScrnMode;MsgAdj;DefTitle;CancelMsg)
Global(SaveScrn;HotElement)

Global(NmeAttrib;ChkAttrib;TstLft;TstRgt)

DefTitle="Coach" // Main title name
ScrnMode=?ScreenHigh-24
CancelMsg=1

Use("wpchlibr.wpm") // Shared library file
ColorScheme()
ScreenSetup()

MsgAdj=DlgMessage()



//*********************************************
//
// MAIN PROGRAM: Coach
//
//*********************************************
Error(Off!)
Cancel(Off!)

Progress=0
If(Not Exists(StndAlone))

Call(OverviewDlg)
If(Result = -1)

QuitCoach()
Endif

Endif
If(var[1] = 1)

Call(PositionCursor1)
Progress=1

Else
If(var[2] = 1)

Call(PositionCursor2)
Endif

Endif
WindowMaximize // Maximize window
DisplayRewrite

Message(&tbox;1005) // Congratulations
Result=MessageBox(DefTitle;&tbox;6)
QuitCoach()

//*********************************************
// END MAIN PROGRAM
//*********************************************



//*********************************************
//
// SUBROUTINE: OverviewDlg
//
//*********************************************
Label(OverviewDlg)

Message(&tbox;1007)
w=65 h=22
DlgCreate(Result;DefTitle;DlgNoOK!+DlgNoCancel!+DlgExit!;;;w;h)

DlgControl(CtrlPushButton!;;"Quit...[Bold On]Esc[Bold Off]";StyCancel!;w-15;h-2)
ForNext(x;1;tbox[0])

DlgControl(CtrlLabel!;;tbox[x];;2;x;w-1;1)
Endfor
DlgControl(CtrlRadioButton!;var[1];"Indent the first line";;3;9;w-1;1)
DlgControl(CtrlLabel!;;"of a paragraph.";;10;10;w-1;1)
DlgControl(CtrlRadioButton!;var[2];"Indent all lines of";;3;15;w-1;1)
DlgControl(CtrlLabel!;;"a paragraph.";;10;16;w-1;1)

var[1]=0 var[2]=0 
DlgEnd

Return



//*********************************************
//
// SUBROUTINE: PositionCursor1
//
//*********************************************
Label(PositionCursor1)

WindowMaximize // Maximize window
Message(&tbox;1302) // Position cursor
FakeIt(DefTitle;&tbox;3;1;0)

Message(&tbox;1008)
Hint=Hint(&tbox;-99;-99;0)

chkcode={ -1;}
keypress={-8154;} // F7 key

Key=DoEdit(1;&chkcode;&keypress)
Call(InsertTab)

Return



//*********************************************
//
// SUBROUTINE: PositionCursor2
//
//*********************************************
Label(PositionCursor2)

Message(&tbox;1303) // Position cursor
FakeIt(DefTitle;&tbox;3;1;0)

Message(&tbox;1008)
Hint=Hint(&tbox;-99;-99;0)

chkcode={ -1;}
keypress={ -8154;} // F7 key

Key=DoEdit(1;&chkcode;&keypress)
Call(InsertIndent)

Return



//*********************************************
//
// SUBROUTINE: InsertTab
//
//*********************************************
Label(InsertTab)

WindowMaximize // Maximize window
Message(&tbox;1304)
FakeIt(DefTitle;&tbox;8;1;0)

chkcode={ -1;}
keypress={ -8183;} // Tab

Key=DoEdit(1;&chkcode;&keypress)
Return



//*********************************************
//
// SUBROUTINE: InsertIndent
//
//*********************************************
Label(InsertIndent)

Call(IndentOn)
Return



//*********************************************
//
// SUBROUTINE: IndentOn
//
//*********************************************
Label(IndentOn)

WindowMaximize // Maximize window
Message(&tbox;1701) // Choose Font menu

hbox1={ "Select [Bold On]Layout[Bold Off] from the menu bar";
"";

}
hbox2={ "Select [Bold On]Alignment[Bold Off] from the Layout menu";

"";
}

hbox3={ "Select [Bold On]Indent →";
"";

}

LayoutMenu(&valid)
allow={ 1565;

}
exallow={0;

}

accel={ 97;
}

exaccel={1;
}

subval={ 1565;
1666;
1567;
1568;
1569;
1570;
1571;
1572;

}
subacl={ 105;

}
exsubacl={ 0;

}
subs={ 1564;

}
sbox={ "Choose [Bold On]Indent →";

}

funkey={ -8157;
}

Repeat
DisplayRewrite
Result=Postit(DefTitle;&tbox;0;9;0;2;0)
MenuActive=0



Show=HotSpot(0;0;62;0;-537;&funkey;1)
Switch(Show)

CaseOf 1: // Show menu
Result=Postit(DefTitle;&tbox;0;9;0;3;0)
ShowMenu("Coach";&sbox;52;6-ScrnMode;3;12;1;0)

CaseOf 1565:
Result=Postit("Coach";&sbox;9;5-ScrnMode;0;0;0)

CaseOf -537: // Accel menu
Result=Postit(DefTitle;&tbox;0;9;0;3;0)
PullDownDlg
Type("l")

CaseOf -8157: // Function key

CaseOf -8085: // Choose menu
Result=Postit(DefTitle;&tbox;0;9;0;3;0)
sbox={ -1}
MenuActive=MenuBarControl("Coach";&hbox1;&sbox;9;5-ScrnMode;108;&funkey)

DEFAULT:
sbox={ -1}
MenuActive=MenuBarControl("Coach";&hbox1;&sbox;9;5-ScrnMode;108;&funkey)

EndSwitch

If(MenuActive <> -1)
If(Show = -8157)

DlgInput(On!)
DisplayRewrite

Indent
Else

If(MenuActive = -8157)
Else

Result=MenuControl(&hbox1;&hbox2;&hbox3;&valid;&allow;&exallow;
&accel;&exaccel;&subval;&subacl;&exsubacl;&subs;&funkey)
MenuActive=Result
If(Result =  1565 or Result = 105)

CancelMenus()
DlgInput(On!)
DisplayRewrite

Indent
Endif

Endif
Endif

Endif
Until(MenuActive <> -1)

Return



//*********************************************
//
// PROCEDURE: Message
//
//*********************************************
PROCEDURE Message(&tbox;Msg)

Switch(Msg)

CaseOf 1005:
tbox={ "";

"This concludes the coach on indenting.";
"";
}

CaseOf 1007:
tbox={ "";

"                 [Bold On]Indenting Lines and Paragraphs[Bold Off]";
"";
"This coach can save you time and frustration by showing you";
"the prefered features for indenting lines and paragraphs.";
"";
"                              ┌────────────────────────────┐";
"                              │                            │";
"                              │      ▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄   │";
"                              │   ▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄   │";
"                              │   ▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄   │";
"                              │   ▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄   │";
"                              │                            │";
"                              │     ▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄   │";
"                              │     ▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄   │";
"                              │     ▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄   │";
"                              │     ▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄▄   │";
"";
}

CaseOf 1008:
tbox={ "";

"";
"Never use spaces to align or indent text.";
"You can use tabs, indents, columns or tables,";
"but never use spaces.";
"";
"The main reason for this is that spaces may not";
"all be equal in width and although your text may";
"appear aligned on the screen, it may appear";
"ragged when printed.";
"";
}

CaseOf 1302:
tbox={ "To indent the first line of a paragraph,";

"";
"▸ Position your cursor where you want to indent";



"";
}

CaseOf 1303:
tbox={ "To indent all lines of a paragraph,";

"";
"▸ Position your cursor where you want to indent";
"";

}

CaseOf 1304:
tbox={ "";

"▸ Press the Tab key to indent the first line of your paragraph.";
"";

}

CaseOf 1701:
tbox={ "";

"To indent all lines of a";
"paragraph,";
"";
"▸ Choose [Bold On]Layout[Bold Off],";
"  then [Bold On]Alignment[Bold Off],";
"  then [Bold On]Indent →[Bold Off].";
"   or";
"▸ Press [Bold On]F4[Bold Off]";
"";

}

DEFAULT:
tbox={ "";

"No message found!";
"";

}

EndSwitch
ENDPROC


